Dateiverknüpfungen per VBA erstellen
Versionen: Excel 2007, 2003, 2002/XP, 2000 und 97

 FRAGE:  Per VBA kann ich mit der Funktion "MkDir" einen Ordner erstellen. Gibt es eine ähnliche Funktion auch für Dateiverknüpfungen, so dass ich zum Beispiel auf dem Windows-Desktop eine Verknüpfung zu einer Excel-Datei einrichten kann?
G. Rienzner

 ANTWORT:  Eine eigenständige Funktion - so wie "MkDir" - gibt es nicht. Eine selbst entwickelte VBA-Prozedur kann das Erstellen einer Dateiverknüpfung aber übernehmen. Dazu wechseln Sie mit Alt+F11 in den Visual Basic-Editor und legen per Einfügen-Modul ein neues Codefenster an. Geben Sie darin folgende Prozedur ein:

Sub CreateShortcut()
Dim objShell As Object
Dim objFolder As Object
Dim objFItem As Object
Dim objScriptHost As Object
Dim objShortcut As Object
Dim strShortcutPath As String
Dim strShortcutName As String
Dim strXLFilename As String

  On Error Resume Next
  Set objShell = CreateObject("Shell.Application")
  Set objFolder = _
      objShell.BrowseForFolder _
      (0, "Wo soll die Verknüpfung erstellt werden?", 0)
  If objFolder Is Nothing Then Exit Sub
  On Error GoTo CrSh_Error
  Set objFItem = objFolder.Self
  strShortcutPath = objFItem.Path
  If ActiveWorkbook.Path <> "" Then
    Set objScriptHost = CreateObject("Wscript.Shell")
    strXLFilename = ActiveWorkbook.Name
    strShortcutName = _
        Left(strXLFilename, _
        InStr(1, strXLFilename, ".xl"))
    Set objShortcut = _
        objScriptHost.CreateShortcut _
        (strShortcutPath & "\\" & _
        strShortcutName & ".lnk")
    With objShortcut
      .TargetPath = ActiveWorkbook.FullName
      .Save
    End With
    MsgBox "Verknüpfung wurde erstellt."
  Else
    MsgBox "Sie müssen die Arbeitsmappe speichern." & _
        vbCr & "Aktion wird abgebrochen..."
  End If
CrSh_End:
  Set objShortcut = Nothing
  Set objFItem = Nothing
  Set objFolder = Nothing
  Set objShell = Nothing
  Set objScriptHost = Nothing
  Exit Sub

CrSh_Error:
  MsgBox "Fehler beim Erstellen der Verknüpfung"
  Err.Clear
  Resume CrSh_End
End Sub

Die Prozedur legt für die aktuelle Arbeitsmappe eine Verknüpfung in einem Ordner Ihrer Wahl an. Die Ordnerauswahl erfolgt per Dialogfenster.

Um die Aktion auszuführen, öffnen Sie per Alt + F8 das Dialogfenster mit der Liste derzeit verfügbarer Makros. Darin doppelklicken Sie auf den Makronamen "CreateShortcut". Anschließend wählen Sie den Zielordner für die Verknüpfung aus. Der Rest läuft automatisch im Hintergrund ab. Später können Sie die aktuelle Arbeitsmappe dann über die Verknüpfung im ausgewählten Zielordner sofort öffnen.

Die Prozedur greift zunächst auf die Windows-Shell zurück, um mit der "BrowseForFolder"-Methode das Dialogfenster für die Ordnerauswahl zu öffnen. Sofern Sie einen Ordner ausgewählt haben, liefert die Methode das entsprechende "Folder"-Objekt. Um an den vollständigen Pfad dieses Ordners zu gelangen, müssen Sie auf den Ordner selbst als "FolderItem" Bezug nehmen. Zu diesem Objekt gehört dann auch eine "Path"-Eigenschaft, die den gesuchten Pfad liefert.

Im nächsten Prozedurblock arbeiten Sie mit dem "Windows Script Host", der wie die Windows-Shell diverse Betriebssystemfunktionen zur Verfügung stellt - unter anderen auch die Methode "CreateShortcut", mit der Sie eine Dateiverknüpfung erstellen. Vorher ermitteln Sie noch den Dateinamen der aktuellen Arbeitsmappe und trennen davon die Endung ".xl*" ab, um den reinen Namen als Verknüpfungsnamen verwenden zu können. Zusammen mit dem oben ausgewählten Pfad der Verknüpfung ergibt sich dann der Parameter für die CreateShortcut-Methode.

Im anschließenden With-Block geben Sie noch an, auf welche Datei die Verknüpfung verweisen soll (hier die aktuelle Arbeitsmappe = "ActiveWorkbook.FullName"). Außerdem speichern Sie die neue Dateiverknüpfung.

Die restlichen Prozedurzeilen dienen lediglich zum "Aufräumen" der Objektvariablen und zur Fehlerbehandlung.

